clear
use inequality_data_70s_correct, clear 

*drop years for which we do not have all observations
drop if ref_year==1979


*ensure summation in Total Vehicle Cons works correctly
foreach var of varlist vehicpq vehiccq cartknpq cartkncq cartkupq cartkucq caropq carocq {
replace `var' = 0 if (`var'==. & ref_year>=1980) 
}

*Generate component totals to create different measures of consumption
* food at home
gen fdhome_tot = .
replace fdhome_tot = fdhome/4 if (ref_year<1970)
replace fdhome_tot = d_fdhome/4 if (ref_year>=1970 & ref_year<=1979) 
replace fdhome_tot = fdhomepq2+fdhomecq2 if (ref_year >= 1980)
*transportation 
gen trans_tot = .
replace trans_tot = (publictransother+automobileops+n_tot_vflow1)/4 if(ref_year<1970) 
replace trans_tot = (d_gas + d_othvehic-d_autofin+d_pubtrans+n_tot_vflow1)/4 if (ref_year>=1970 & ref_year<=1979)
replace trans_tot = (transpq+transcq+n_tot_vflow1-(vehfinpq+vehfincq+vehicpq+vehiccq+cartknpq+cartkncq+cartkupq+cartkucq+caropq+carocq)) if (ref_year>=1980)
*utilities 
gen util_tot = .
replace util_tot = utilites/4 if (ref_year<1970) 
replace util_tot = d_utility/4 if (ref_year>=1970 & ref_year<=1979) 
replace util_tot = utilpq+utilcq if (ref_year>=1980) 
*Adjusted utilities from predictions
*replace util_tot = n_adj_util_floor if (cutenure==4 & _merge_rnt==3)
*housing
gen hous_tot = . 
replace hous_tot = (totrentdwe +hflow)/4 if (ref_year<1970) 
replace hous_tot = (d_rent + hflow)/4 if (ref_year>=1970 & ref_year<=1979) 
replace hous_tot = rendwepq+rendwecq+hflow if (ref_year>=1980) 
*difference in adjusted and reported utilities is utility payment included in rent (forced to be positive) 
*gen diff_rnt = n_adj_util_floor - (utilpq+utilcq) if (cutenure==4 & _merge_rnt==3) 
*replace hous_tot = hous_tot - diff_rnt if (cutenure==4 & _merge_rnt==3 & diff_rnt>0 & diff_rnt!=.) 
 
*Generate core less food, non-housing core, and core less food and utilities 
gen double cons10 = ((cons5*scale)-fdhome_tot)/scale 
gen double cons11 = ((cons5*scale)-hous_tot)/scale 
gen double cons12 = ((cons10*scale)-util_tot)/scale
gen double cons13 = ((cons5*scale)-gas_mo)/scale
gen double cons14 = ((cons5*scale)-(n_tot_vflow1/4))/scale if (ref_year<1980)
replace cons14 = ((cons5*scale)-n_tot_vflow1)/scale if (ref_year>=1980)

*Redo 1959-1961 price adjustment 
foreach var of varlist cons* {
replace `var' = `var'*1.028 if yr_id==9
replace `var' = `var'*1.0098 if yr_id==0
}

*Generate cpi defalator
preserve 
clear         
import excel "C:\Users\worri\Dropbox\Poverty\price_indices17_JH.xlsx", sheet("dat_file 2017") firstrow 
rename Year ref_year
rename cpiursadj cpi_u_rs_adj_17
keep ref_year cpi_u_rs_adj_17
replace cpi_u_rs_adj_17 = 1/cpi_u_rs_adj_17
tempfile cpi_u_rs_adj_17
save `cpi_u_rs_adj_17'
restore
merge m:1 ref_year using `cpi_u_rs_adj_17'
drop _merge


*Generate real consumption figures
foreach var of varlist cons* {
gen r_`var' = `var'*cpi_u_rs_adj_17
}

label var r_cons5 "Well measured consumption (core)" 
label var r_cons6 "Total Consumption"
label var r_cons3 "Total Consumption Plus HI"
label var r_cons10 "Core less food" 
label var r_cons11 "Non-housing core"
label var r_cons12 "Core less food and utilities"
label var r_cons13 "Core less gas and motor oil"
label var r_cons14 "Core less vehicle"
label var r_cons1 "Expenditure"
*create binary variable for gender
gen male=2-sex_ref
*Gen family type dummies
forvalues x=1/5 {
gen fam`x'=0
replace fam`x'=1 if ftype==`x'
gen fam_male`x'=fam`x'*male
}
*Create region dummies 
forvalues x=1/4 {
gen reg`x' =0
replace reg`x'=1 if region==`x'
replace reg`x'=0 if bls_urbn==2
}
gen urban=2-bls_urbn
*Create race dummies
gen race1=0
replace race1=1 if (ref_race==1 & hispanic==0)
gen race2=0
replace race2=1 if (ref_race==2 & hispanic==0)
gen race3=1-race1-race2
*Gen employment indicators 
gen emp=0 if (incweek1 != .)
replace emp=1 if (incweek1 !=. & incweek1>0)
gen emp2=0 if (incweek2 != .) 
replace emp2=1 if (incweek2 != . & incweek2>0)
gen emp_type=0
replace emp_type=1 if (emp==1 & emp2==1)
replace emp_type=2 if (emp==1 & emp2==0)
replace emp_type=3 if (emp==0 & emp2==1)
replace emp_type=4 if (emp==0 & emp2==0)
*Gen race indicator  
gen race_type=. 
replace race_type=1 if (race1==1)
replace race_type=2 if (race2==1)
replace race_type=3 if (race3==1)
*Gen education indicators and dummies (note, ed_type exists for earlier years, drop before creating new variable)
rename ed_type old_ed_type
gen ed_type=.
replace ed_type = old_ed_type if (ref_year<1970)
replace ed_type=1 if (ref_year>=1972 & ref_year<=1973 & (educ_ref==1 | educ_ref==2 | educ_ref==6))
replace ed_type=2 if (ref_year>=1972 & ref_year<=1973 & educ_ref==3)
replace ed_type=3 if (ref_year>=1972 & ref_year<=1973 & educ_ref==4)
replace ed_type=4 if (ref_year>=1972 & ref_year<=1973 & educ_ref==5)
replace ed_type=1 if (qyear>=801 & qyear<=955 & (educ_ref==1 | educ_ref==2 | educ_ref==7))
replace ed_type=2 if (qyear>=801 & qyear<=955 & educ_ref==3)
replace ed_type=3 if (qyear>=801 & qyear<=955 & educ_ref==4)
replace ed_type=4 if (qyear>=801 & qyear<=955 & (educ_ref==5 | educ_ref==6))
replace ed_type=1 if (qyear>955 & (educ_ref==0 | educ_ref==10 | educ_ref==11))
replace ed_type=2 if (qyear>955 & educ_ref==12)
replace ed_type=3 if (qyear>955 & (educ_ref>=13 & educ_ref<=14))
replace ed_type=4 if (qyear>955 & (educ_ref>=15 & educ_ref<=17))
forvalues x=1/4 {
gen ed`x'=0
replace ed`x'=1 if ed_type==`x'
}
*Gen age indicators 
gen age_ref_2=age_ref^2
gen age_ref_3=age_ref^3

* Fix "race1"
replace race1=0 if ref_year<=1973
replace race1=1 if ref_race==1 & ref_year<=1973
keep if ref_year==1972|ref_year==1990|ref_year==2000|ref_year==2017
keep newid ref_year wgt20 r_cons5 r_cons10 age_ref age_ref_2 age_ref_3 fam2 fam3 fam4 fam5 race1 ed2 ed3 ed4 
save decomp_master, replace


********************************************************************************
clear
use inequality_data_for_60.dta, clear 
*Gen family type dummies
forvalues x=1/5 {
gen fam`x'=0
replace fam`x'=1 if ftype==`x'
}
*Gen education dummies 
forvalues x=1/4 {
gen ed`x'=0
replace ed`x'=1 if ed_type==`x'
}
*Gen age indicators 
gen age_ref_2=age_ref^2
gen age_ref_3=age_ref^3

* Fix "race1"
replace race1=0 if ref_year==1961
replace race1=1 if ref_race==1 & ref_year==1961

keep if ref_year==1961
keep newid ref_year wgt20 r_cons5 r_cons10 age_ref age_ref_2 age_ref_3 fam2 fam3 fam4 fam5 race1 ed2 ed3 ed4 
save decomp_60, replace

********************************************************************************
clear
use inequality_data_for_80.dta, clear 
*Gen family type dummies
forvalues x=1/5 {
gen fam`x'=0
replace fam`x'=1 if ftype==`x'
}
*Gen education dummies 
forvalues x=1/4 {
gen ed`x'=0
replace ed`x'=1 if ed_type==`x'
}
*Gen age indicators 
gen age_ref_2=age_ref^2
gen age_ref_3=age_ref^3

keep if ref_year==1980
keep newid ref_year wgt20 r_cons5 r_cons10 age_ref age_ref_2 age_ref_3 fam2 fam3 fam4 fam5 race1 ed2 ed3 ed4 
save decomp_80, replace
